Modifying a structural equation model of child dietary intake using the Lagrange Multiplier test in SAS® PROC CALIS for MWSUG 2014

Lauren Cook, University of Southern California, Los Angeles, CA
Chih-Ping Chou, University of Southern California, Los Angeles, CA
Nicole Gatto, MPH, PhD, Loma Linda University, Loma Linda, CA
Jaimie N Davis, RD, PhD, University of Texas, Austin, TX
Donna Spruijt-Metz, MFA, PhD, University of Southern California, Los Angeles, CA

ABSTRACT

Structural equation models (SEM) allow for simultaneous evaluation of causal pathways between multiple predictors and outcomes, including latent variables, given a hypothesized theoretical model. If an initial SEM does not meet appropriate fit criteria, model modification can achieved through utilization of statistical results from the Lagrange Multiplier (LM) test and theoretical knowledge.

To demonstrate modification of a SEM, baseline data from a garden-based nutrition intervention with elementary school children were examined using SAS® PROC CALIS. The two outcome variables of interest are fruit and vegetable intake (individually). Several measured determinants are hypothesized to predict these practices, including two latent variables, willingness to try fruit and vegetables, which have six indicator variables each.

Initial model fit was not acceptable ($\chi^2$: 407.9, 111 df, p<0.0001; RMSEA: 0.09, CFI: 0.89). From the LM statistic we identified several unaccounted for correlations between errors on indicators of willingness to try fruits and vegetables. These correlations are theoretically sensible given that items on these scales are worded identically, save for the words “fruit” and “vegetable”, and these modifications were made. Once correlations between comparable indicators were added to the SEM, model fit was improved ($\chi^2$: 234.0, 105 df, p<0.0001; RMSEA: 0.06, CFI: 0.95). Additional relationships identified by the LM test were evaluated, but none were theoretically meaningful, and the second model was accepted as final.

Use of the LM test in PROC CALIS facilitates theoretically appropriate modification of an a priori SEM in order to improve overall model fit and produce more reliable parameter estimates.

INTRODUCTION

Structural equation modeling (SEM) is an analytical approach that allows for exploration of complex multivariate relationships. There are two primary benefits of this technique: 1) latent factors can be observed from knowledge of indicator variables, as with factor analysis, and 2) multiple relationships between independent and dependent variables can be simultaneously observed.

As with general linear regression, checks for model validity are an essential part of the modeling process. Unlike linear regression, statistical tests are performed to ensure the model meets criteria for acceptable fit. When overall model fit does not meet standards, the Lagrange Multiplier (LM) test can be employed to identify parameters that can be added to the model (either directional pathways or covariances between variables, factors or error terms) in order to improve fit. However, given its post-hoc nature, the LM test is a somewhat controversial approach, so must be used in conjunction with theoretical relevance.

This paper will demonstrate the use of the LM test to improve model fit in SEM using PROC CALIS. The specific application will be in exploring the determinants of fruit and vegetable intake and preferences in elementary school children, as part of a nutrition, cooking and gardening intervention for obesity prevention.  

VARIABLES

One structural model will include variables specific to fruits, and variables specific to vegetables (all data are cross-sectional). The model is structured in this way because fruits and vegetables have varying determinants, and thus the strength of relationships may differ between predictors of fruit versus vegetables intake and preferences. Also, there may be some effects of fruit–related variables on vegetable-related variables, and vice versa. The initial hypothesized model is included in Figure 1.
Figure 1: Initial hypothesized model

LIST OF VARIABLES

**Fruit intake** and **vegetable intake**: These variables are measured via the Block Kids Food Screener (last week version), and intake of each of these food types is provided in cup equivalents.

**Fruit identification** and **vegetable identification**: Students were given a list of 8 fruit items and 17 vegetable items, and were asked if they knew these items, or not. A standardized sum (accounting for unanswered questions) was obtained, with a range of 6-8 items answered about fruits, and 11-17 items answered about vegetables.

**Fruit preferences** and **vegetable preferences**: For each item identified from the above question, students were asked if they liked this food “A lot”, “A little”, or “Not at all”. Mean scores were used for fruit items and vegetable items (factor analysis was not used due to the large number of indicators).

**Willingness to try fruit** and **willingness to try vegetables**: The willingness to try scale has six items each for fruits and vegetables, with questions such as “How much do you like tasting new vegetables?” and responses ranging from “Not at all” to “A lot”. The variables for the six items in each scale (specific to fruits or vegetables) served as indicators for a latent factor, either willingness to try fruits or willingness to try vegetables, respectively.

DATA ANALYSIS

DATA PREPARATION

Data appeared to be missing at random, so to account for missingness, the correlation matrix was analyzed instead of raw data. The following code was used to perform this task:

```r
proc corr data=las2 out=las_cor nosimple noprob;
var sex ethnicity bmiz age fruit_intake veg_intake fruit_id
veg_id veg_willing_1 veg_willing_2 veg_willing_3 veg_willing_4 veg_willing_5 veg_willing_6 fruit_willing_1 fruit_willing_2 fruit_willing_3 fruit_willing_4 fruit_willing_5 fruit_willing_6 fruit_pref veg_pref;
run;
```

MODEL 1 SPECIFICATION

The code below was used to fit the initial hypothesized model. In the PROC line we specify that the data form is a correlation matrix, specify that we would like to use maximum likelihood estimation, and the ‘mod’ command indicates we would also like to output the results of the LM test.

Following the ‘lineqs’ statement, each equation represents a path we would like to include in the model. Terms such as ‘b1’ give a name to the parameter to reference in the output, and each equation also includes an error term. The terms ‘f1’ and ‘f2’ refer to our two latent factors, willingness to try fruits and willingness to try vegetables, and note
these must be indicated in this form (rather than creating a meaningful factor name, such as ‘fruit_willingness_factor’; it is acceptable in CALIS to use original variable names rather than renaming as ‘v1’, etc. for those included in the database).

In the ‘variance’ statement (or ‘std’, which performs the same function), we indicate the parameters of which we want to estimate the variance. Similarly in the ‘cov’ statement, we indicate the relationships of which we would like to estimate the covariances, and we indicate a name for this parameter. We can name parameters in the ‘variance’ statement, as well, but this is not necessary. Note that for dependent variables (fruit and vegetable preferences and intake), we estimate the covariances of the error terms, not the variables themselves.

```
proc calis corr data=las_corr method=ml mod;
lineqs
  fruit_pref = b1 fruit_id + b2 veg_id + b3 f1 + b4 f2 + e1, 
  veg_pref = b5 fruit_id + b6 veg_id + b7 f1 + b8 f2 + e2, 
  fruit_willing_1 = a1 f1 + e10, 
  fruit_willing_2 = a2 f1 + e11, 
  fruit_willing_3 = a3 f1 + e12, 
  fruit_willing_4 = a4 f1 + e13, 
  fruit_willing_5 = a5 f1 + e14, 
  fruit_willing_6 = a6 f1 + e15, 
  veg_willing_1 = a7 f2 + e16, 
  veg_willing_2 = a8 f2 + e17, 
  veg_willing_3 = a9 f2 + e18, 
  veg_willing_4 = a10 f2 + e19, 
  veg_willing_5 = a11 f2 + e20, 
  veg_willing_6 = a12 f2 + e21, 
fruit_intake = b9 fruit_id + b10 veg_id + b11 f1 + b12 f2
              + b13 fruit_pref + b14 veg_pref + e3, 
veg_intake = b15 fruit_id + b16 veg_id + b17 f1 + b18 f2
              + b19 fruit_pref + b20 veg_pref + e4;
variance
  e1, e2, e3, e4, e10, e11, e12, e13, e14, 
e15, e16, e17, e18, e19, e20, e21;
  cov
  fruit_id veg_id= theta1, 
  f1 f2 = theta2, 
  fruit_id f1 = theta3, 
  veg_id f2 = theta4, 
  fruit_id f2 = theta5, 
  veg_id f1 = theta6, 
  e1 e2 = theta7, 
  e3 e4 = theta8;
run;
```

**MODEL 1 OUTPUT**

The following output summarizes the fit of this model. Fit of this model is not acceptable (we would like to see a value >0.95 for the Bentler Comparative Fit Index (CFI), and a value <0.05 for the Root Mean Square Error of Approximation (RMSEA) Index). Therefore, we look to the LM test results.
Results of the LM test are included below. The first two tables indicate additional pathways that can be created between observed variables (distinguishing them such that the first table includes both ‘predictor’ and ‘outcome’ variables, and the second table only includes ‘predictor’ variables (ie, identification and willingness variables only); technically our willingness indicators are dependent on the willingness factor), and ranks them by magnitude of change to model fit. However, inclusion of any of these pathways indicated in the first two tables is not desirable. Since we hypothesize a relationship between the willingness to try factors and preferences and intake, it would not make sense to include pathways in the model directly linking the factor indicators with those outcome variables.

The third table shows additional covariances between errors that could be added to the model, again ranked by magnitude. From examining these, we see that some of these pathways make sense. For example, ‘e21’ and ‘e15’ both refer to the sixth item on the willingness to try scales, and these questions are worded identically, with the exception of the words ‘fruits’ or ‘vegetables’. The same is true for ‘e10’ and ‘e16’, and ‘e20’ and ‘e14’.

Display 1: Selected output from first structural equation model (fit statistics)
Display 2: Selected output from first structural equation model (Lagrange Multiplier test)

MODEL 2 SPECIFICATION

To modify the model, we add six additional covariance parameters to the model, such that correlations are included between comparable items on the fruit and vegetable willingness to try subscales (Figure 2).
The revised code is below, with the additional lines to the ‘cov’ statement being the only modifications made.

```
proc calis corr data=las_corr method=ml mod;
lineqs
fruit_pref = b1 fruit_id + b2 veg_id + b3 f1 + b4 f2 + e1,
veg_pref = b5 fruit_id + b6 veg_id + b7 f1 + b8 f2 + e2,
fruit_willing_1 = a1 f1 + e10,
fruit_willing_2 = a2 f1 + e11,
fruit_willing_3 = a3 f1 + e12,
fruit_willing_4 = a4 f1 + e13,
fruit_willing_5 = a5 f1 + e14,
fruit_willing_6 = a6 f1 + e15,
veg_willing_1 = a7 f2 + e16,
veg_willing_2 = a8 f2 + e17,
veg_willing_3 = a9 f2 + e18,
veg_willing_4 = a10 f2 + e19,
veg_willing_5 = a11 f2 + e20,
veg_willing_6 = a12 f2 + e21,
fruit_intake = b9 fruit_id + b10 veg_id + b11 f1
  + b12 f2 + b13 fruit_pref + b14 veg_pref + e3,
veg_intake = b15 fruit_id + b16 veg_id + b17 f1
  + b18 f2 + b19 fruit_pref + b20 veg_pref + e4;
```

```
variance
e1, e2, e3, e4, e10, e11, e12, e13, e14,
e15, e16, e17, e18, e19, e20, e21;
```

```
cov
fruit_id veg_id= theta1,
f1 f2 = theta2,
fruit_id f1 = theta3,
veg_id f2 = theta4,
fruit_id f2 = theta5,
veg_id f1 = theta6,
e1 e2 = theta7,
e3 e4 = theta8,
e10 e16 = theta9,
e15 e21 = theta10,
e13 e19 = theta11,
```
MODEL 2 OUTPUT

This code gives us the following output for fit statistics. We see that the model fit is improved, such that the CFI is within an acceptable range (0.95), and the RMSEA value is borderline (0.059). Also the Chi-square value is improved from 407.9 to 234.0 (with six fewer degrees of freedom), although still with a p-value >0.05. Results of the LM test were again evaluated, but none were theoretically meaningful, so we consider this model acceptable to move forward with hypothesis testing.

Display 3: Selected output from second structural equation model (fit statistics)
The output below indicates the standardized estimates for the linear equations in the model, and the t-value indicates the significance of these associations. We see that fruit identification and willingness to try fruit are significant predictors of fruit preferences, and that willingness to try vegetables is a significant predictor of vegetable preferences. For fruit intake, only fruit identification is a significant predictor; and for vegetable intake, fruit identification, vegetables identification, and willingness to try vegetables are significant predictors.

### Standardized Results for Linear Equations

<table>
<thead>
<tr>
<th></th>
<th>Estimate</th>
<th>Std. Error</th>
<th>t Value</th>
<th>p Value</th>
</tr>
</thead>
<tbody>
<tr>
<td>fruit_willing_1</td>
<td>0.569</td>
<td>0.017</td>
<td>14.257</td>
<td>&lt;.0001</td>
</tr>
<tr>
<td>fruit_willing_2</td>
<td>0.528</td>
<td>0.016</td>
<td>16.348</td>
<td>&lt;.0001</td>
</tr>
<tr>
<td>fruit_willing_3</td>
<td>0.576</td>
<td>0.016</td>
<td>17.315</td>
<td>&lt;.0001</td>
</tr>
<tr>
<td>fruit_willing_4</td>
<td>0.610</td>
<td>0.018</td>
<td>18.644</td>
<td>&lt;.0001</td>
</tr>
<tr>
<td>fruit_willing_5</td>
<td>0.609</td>
<td>0.019</td>
<td>19.442</td>
<td>&lt;.0001</td>
</tr>
</tbody>
</table>

**Display 4:** Selected output from second structural equation model (linear equations)
CONCLUSION

These results indicate that identification and willingness to try fruits and vegetables are predictors of preferences and intake, but preference is not predictive of intake, as has previously been demonstrated in the literature. A structural equation model is a useful way to examine these data because preferences can be evaluated as a mediator between intake and the predictors identification and willingness to try, but we see from these results that it is not one.

When specifying the model, the LM test is a helpful tool for model modification, to ensure that overall model fit is appropriate and that parameter estimates are reliable. However, prudence must be exercised when employing this approach so that models remain theoretically sound. Just as inappropriate model fit can diminish findings, so can too an illogical model.

REFERENCES


ACKNOWLEDGMENTS

This project is funded by NIH R21 DK094066-02.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Lauren Cook
Enterprise: University of Southern California
Address: 2250 Alcazar St, CSC-200
City, State ZIP: Los Angeles, CA, 90089
Work Phone: 323-442-2735
Fax: 323-442-4103
E-mail: laurenco@usc.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Other brand and product names are trademarks of their respective companies.